
Группа Пример к теме
4 Запуск звезда-треугольник

Краткое описание

Данный пример управляет запуском звезда-треугольник асинхронного двигателя.
Двигатель запускается после нажатия кнопки ВКЛ на входе E0.0 в режиме звезды. По
прошествии установленного времени 5сек двигатель переключается в режим треугольник.

Если нажата кнопка ВЫКЛ или сработал автомат защиты двигателя, находящиеся на
входах E0.1 и E0.2, двигатель выключается. Если одновременно нажаты кнопки ВЫКЛ и
ВКЛ, то двигатель остается выключенным.

Структура программы

ВЫКЛ

ЗвездаТреугольник

E0.0

E0.1 / E0.2

5 сек

Описание программы вкл. листинг

В начале цикла программа проверяет, должен ли быть установлен вспомогательный
меркер M10.0 для блокировки. Это будет сделано, если Кнопки ВЫКЛ (E0.1) и ВКЛ (E0.0)
нажаты одновременно. Блокировка будет снята только, если обе кнопки снова находятся
в исходном состоянии. Блокировка служит для защиты от неверных условий.

Для процесса включения имеется вспомогательный меркер M11.0. Он будет установлен
при включении, т.е. если кнопка на входе E0.0 замкнута и переключатель сети еще не
сработал. Он будет также установлен, если двигатель находится в режиме звезды, т.е.
переключатель сети (A0.0) и переключатель звезды (A0.1) активны. Установка
вспомогательного меркера M11.0 происходит  впрочем только тогда, если не разомкнуты
автомат защиты двигателя (E0.2) или кнопка ВЫКЛ (E0.1), а переключатель треугольника
(A0.2) неактивен.

Выход A0.1 переключателя звезды будет установлен, если установлен вспомогательный
меркер запуска M11.0, отсутствует блокировка (M10.0) и время переключения (5 сек
таймера T37) еще не закончилось.
Время переключения запускается в момент установки вспомогательного меркера запуска
M11.0, в случае если отсутствует блокировка. Временной базис таймера T37 - 100 мсек,
т.е. при загрузке значения 50 время переключения составит 5 сек.

Выход A0.0 переключателя сети устанавливается, если или кнопка ВКЛ на E0.0, выход
A0.1 переключателя звезды или переключатель сети уже активны и нет команды ВЫКЛ от
кнопки на E0.1 или сработала защита двигателя на E0.2 и отсутствует блокировка.

Переключатель треугольника, т.е. выход A0.2, устанавливается при активном
переключателе сети и уже неактивном переключателе звезды.

Размер программы составляет 40 слов.

KOP (S7-MicroDOS) AWL (IEC)

Основная программа

// TITEL =  ЗАПУСК ЗВЕЗДА-ТРЕУГОЛЬНИК БЕЗ ОТВЕТНОГО СООБЩЕНИЯ

// E0.0 Кнопка вкл.
// E0.1 Кнопка выкл. Нормально замкнута
// E0.2 Защита двигателя Нормально замкнута

// A0.0 Переключатель сети
// A0.1 Переключатель звезды
// A0.2 Переключатель треугольника
// T37 Таймер переключения 5 сек

// БЛОКИРОВКА

│ E0.1 E0.0 M10.0 K1
1 ├─┤ / ├───┤ ├───────────────────────────────────────────( S )

│

LDN  E0.1 // Кнопка Выкл нажата
U  E0.0 // Кнопка Вкл нажата
S  M10.0 ,1 // Вспомогательный меркер для блокировки

│ E0.1 E0.0 M10.0 K1
2 ├─┤ ├───┤ / ├───────────────────────────────────────────( R )

│

LD  E0.1 // Кнопка Выкл не нажата
UN  E0.0 // Кнопка Вкл не нажата
R  M10.0 ,1 // Блокировка снята

// ВКЛЮЧЕНИЕ

│ E0.0 A0.0 E0.2 E0.1 A0.2 M11.0
3 ├─┤ ├───┤ / ├─┬─┤ ├───┤ ├───┤ / ├───────────────────( )

│ │
│ A0.0 A0.1 │
├─┤ ├───┤ ├─┘
│

LD  E0.0 // Кнопка Вкл нажата
UN  A0.0 // Нет переключателя сети
LD  A0.0 // Переключатель сети
U  A0.1 // Переключатель звезды
OLD 

LD  E0.2 // Защита двигателя OK
U  E0.1 // Кнопка ВЫКЛ не нажата
UN  A0.2 // Нет переключателя треугольника
ULD 
=  M11.0 // Вспомогательный меркер

// ВКЛЮЧЕНИЕ ПЕРЕКЛЮЧАТЕЛЯ ЗВЕЗДЫ

│ M11.0 M10.0 T37 A0.1
4 ├─┤ ├───┤ / ├───┤ / ├───────────────────────────────────( )

│

LD  M11.0 // Вспомогательный меркер
UN  M10.0 // Нет блокировки
UN  T37 // Время переключения не закончилось
=  A0.1 // Переключатель звезды
 

// ЗАПУСК ТАЙМЕРА ПЕРЕКЛЮЧЕНИЯ

│ M10.0 M11.0 TON───T37
5 ├─┤ / ├───┤ ├───────────────────────────────────────────┤IN │

│ │ │
│ K50┤PT │
│ │ │
│ └───────┘

LDN  M10.0 // Нет блокировки
U  M11.0 // Вспомогательный меркер
TON  T37,50 // Запуск таймера переключения (5 сек)

// ВКЛЮЧЕНИЕ ПЕРЕКЛЮЧАТЕЛЯ СЕТИ

│ E0.1 E0.2 M10.0 E0.0 A0.1 A0.0
6 ├─┤ ├───┤ ├───┤ / ├─┬─┤ ├───┤ ├─┬─────────────────( )

│ │ │
│ │ A0.0 │
│ └─┤ ├─────────┘
│

LD  E0.1 // Кнопка ВЫКЛ не нажата
U  E0.2 // Защита двигателя OK
UN  M10.0 // Нет блокировки
LD  E0.0 // Кнопка Вкл нажата
U  A0.1 // Переключатель звезды
O  A0.0 // Переключатель сети
ULD 
=  A0.0 // Переключатель сети

// ВКЛЮЧЕНИЕ ПЕРЕКЛЮЧАТЕЛЯ ТРЕУГОЛЬНИКА

│ A0.0 A0.1 A0.2
7 ├─┤ ├───┤ / ├───────────────────────────────────────────( )

│

LD  A0.0 // Переключатель сети
UN  A0.1 // Переключатель звезды
=  A0.2 // Переключатель треугольника

│
8 ├─────────────────────────────────────────────────────────( MEND )

│

MEND // Конец основной программы

Указания по преобразованию

Для того чтобы преобразовать TOOLITE2 AWL в S7-Micro/DOS AWL

            �  Установите ‘K’ перед каждым числом, не являющимся 16-ричной
      константой (напр. 4 → K4)

            �  Замените ‘16#’ на ‘KH’ для всех 16-ричных констант
      (напр. 16#FF → KHFF)

            �  Поставьте запятые для смены полей. Используйте клавиши
                 перемещения или клавишу TAB для перехода от поля к полю.
            �  Для преобразования программы S7-Micro/DOS AWL в KOP-форму нужно

начинать каждый сегмент словом ‘NETWORK’ и номером. Каждый сегмент в
этом примере имеет свой номер на диаграмме KOP. Используйте NWENFG в
меню редактора для ввода нового сегмента. Команды MEND, RET, RETI, LBL,
SBR и INT требуют отдельных сегментов.

Общие указания

Примеры SIMATIC S7-200 предоставляются заказчику бесплатно. Данные примеры не
привязаны к конкретной задаче и являются общей информацией о возможностях
применения S7-200. Решение заказчика может отличаться от приведённого здесь.

За правильную работу системы заказчик несёт ответственность сам. Мы обращаем Ваше
внимание на действующие нормы Вашей страны и предписания по установке
соответствующей системы. Ошибки и изменения возможны.




